5321 Industrial Organization: Seminar 2

Marcus Hagman*
Stockholm School of Economics

May 2024

This Exercise

As economists we are often interested in simulating policy counterfactuals. For example:

- Labor: What happens to employment if the minimum wage is raised?
- Macro: What happens to growth if the pension age is raised?
- **IO: What happens to consumer prices if a merger goes through?**

In the absence of natural experiments (or sometimes as a complement), economists often employ *structural models*. This approach consists of first setting up a theoretical framework in which the answer to the question is well defined. Then we estimate the parameters of the model. Once we have done that, we can change some aspect of the decision makers' environment to see how outcomes of interest change.

In the context of merger simulation, the model primitives to be estimated are demand elasticities and marginal costs. These are generally assumed to be observed by the firm but not by the econometrician. However, given the right conditions, they can be estimated:

- Given enough structure on demand (a model of utility maximizing behavior) and exogenous variation in price, demand elasticities can be estimated.
- Given demand elasticities and profit maximizing behavior of firms (Nash-Bertrand), marginal costs can be estimated.

Once these primitives have been estimated, we can predict what would happen to prices if a merger goes through, which changes the profit functions of firms.

In this exercise, we will simulate data. Simulated data has the advantage that we can know what the true counterfactual is. That is, we can find what would actually happen to prices

*marcus.hagman@phdstudent.hhs.se

in the context of the model if a merger goes through. Then we can limit ourselves to only observing prices, quantities, locations of production sites and locations of markets. Using only these observables, we will see how closely we can estimate the price effect of a merger.

Model Set Up

A country consists of 2000 towns, each one located along a straight road that runs West to East. The towns are located 1 km away from each other. Three firms produce goods, and their production sites are located 0, 500 and 1300 km respectively from the Westernmost point of the road. The marginal cost of selling a good in a given town is an increasing function of the town's distance from the production site. A consumer i in town t gets the following utility from consuming good j:

$$u_{ijt} = \gamma_j - \alpha p_{jt} + \xi_{jt} + \epsilon_{ijt}$$

Where:

- γ_j denotes the intrinsic quality of the product.
- α denotes sensitivity to price.
- ξ_{jt} denotes the town-specific preferences for a particular product. We assume this is a random variable which is iid across towns and products (strong assumption).
- ϵ_{ijt} is an individual specific taste shock. Distributed iid EV Type 1.

Moreover the mean utility for the ouside option is zero:

$$u_{i0t} = \epsilon_{i0t}$$

Each consumer will choose exactly one option according to which yields the most utility. Each market has sufficiently many consumers that we need only consider the large sample properties of choice probabilities.

The main question we are interested in answering is: **What would happen to prices if firm 1 merged with firm 3?**

Generating the Data

All code is written in Stata. We start by generating all variables except for prices.

```
clear all
set seed 12345

cd "C:\Users\marcu\Dropbox\Demand Est\Seminar 2"

* Set the number of observations
set obs 6000

* Create market, product, firm identifiers
gen market = ceil(_n/3)
gen product = 1 + mod(_n - 1, 3)
gen firm = product

* Generate xi from a normal distribution N
gen xi = rnormal(0, 2)

* Generate distance to production site
gen dist_to_production = 0
replace dist_to_production = abs(market - 0) if product == 1
replace dist_to_production = abs(market - 500) if product == 2
replace dist_to_production = abs(market - 1300) if product == 3

* Generate marginal cost
gen mc = product + dist_to_production/1000

* Utility parameters
gen gamma = product
gen alpha = 1.5
```

A firm f which owns the set of products F sets its price in each market to maximize profits.

$$\pi_{ft}(p_t) = \sum_{j \in F} (p_{jt} - c_{jt}) s_{jt}(p_t)$$

Then the following FOC needs to hold for every j in F:

$$\frac{\partial \pi_{ft}(p_t)}{\partial p_{jt}} = s_{jt}(p_t) + \sum_{j' \in F} (p_{j't} - c_{j't}) \frac{\partial s_{j't}(p_t)}{\partial p_{jt}} = 0 \qquad \forall j \in F$$

Recall from the slides that under the structure of utility conditions outlined, we get the demand functions:

$$s_{jt} = \frac{\exp(\gamma_j - \alpha p_{jt} + \xi_{jt})}{1 + \sum_{j'} \exp(\gamma_{j'} - \alpha p_{j't} + \xi_{j't})}$$

Which has the following slopes with respect to prices:

$$\frac{\partial s_{j't}(p_t)}{\partial p_{jt}} = \begin{cases} -\alpha s_j(1 - s_j) & \text{if } j = j' \\ \alpha s_{j'} s_j & \text{if } j \neq j' \end{cases}$$

The equilibrium price vector in Stata needs to satisfy all first order conditions. Unfortunately, unlike many of the models you solved for PS1, the FOCs don't simplify to neat closed form expressions. However, there are methods for finding equilibrium objects computationally. The following code provides such an example (the code is bad but works). The approach involves starting to set prices equal to marginal cost, and then adjusting it upward if marginal profits are positive, down if marginal profits are negative.

```
* Code below finds equilibrium prices by first starting at marginal cost and
*looping until convergence.
gen p = mc
bysort market: egen s_denominator = total(exp(gamma - alpha*p + xi))
replace s_denominator = s_denominator + 1
gen s = exp(gamma - alpha*p + xi) / s_denominator
gen marg_profit = (p-mc)*(-s*(1-s)) + s
forvalues i = 1/10000 {
        replace p = p + marg_profit
        drop s_denominator
        bysort market: egen s_denominator = total(exp(gamma - alpha*p + xi))
        replace s_denominator = s_denominator + 1
        replace s = exp(gamma - alpha*p + xi) / s_denominator
        replace marg_profit = s + (p-mc)*(-alpha*s*(1-s))
}
```

I save two .dta files, one of which contains all variables and one which contains only the variables which are visible to the econometrician.

```
* Save data with unobservables
save "1_data\seminar_2_data_unobservable.dta", replace

* Drop variables unobserved by the econometrician
drop xi mc gamma alpha marg_profit

* Save data with only observables
save "1_data\seminar_2_data_observable.dta", replace
```

True Counterfactuals

This section contains the true counterfactuals that we will later try to estimate without unobservables. We start by loading the data with unobservables and calculating average marginal costs for each product.

```
clear all
set seed 12345
cd "C:\Users\marcu\Dropbox\Demand Est\Seminar 2"
use "1_data\seminar_2_data_unobservable.dta"

* Average marginal cost
bysort product: summarize mc
```

```
. bysort product: summarize mc
```

-> product = 1

Variable	Obs	Mean	Std. dev.	Min	Max
mc	2,000	2.0005	.5774946	1.001	3

-> product = 2

Variable	Obs	Mean	Std. dev.	Min	Max
mc	2,000	2.62525	.439309	2	3.5

-> product = 3

Variable	Obs	Mean	Std. dev.	Min	Max
mc	2,000	3.54485	.3552958	3	4.299

The same for Lerner index.

```
* Average Lerner
gen lerner = (p - mc)/p
bysort product: summarize lerner
```

```
. bysort product: summarize lerner
```

-> product = 1

Variable	Obs	Mean	Std. dev.	Min	Max
lerner	2,000	.2887013	.0893361	.1291011	.7481685

-> product = 2

Variable	Obs	Mean	Std. dev.	Min	Max
lerner	2,000	.2266204	.0499647	.0679106	.5411247

-> product = 3

Variable	Obs	Mean	Std. dev.	Min	Max
lerner	2,000	.1735046	.0327092	.0815573	.4850651

Next we calculate what the prices would be if the owners of products 1 and 3 merged using the same computational method as before. The code differs from before in that the first order conditions for the prices of 1 and 3 will now include the effect that one's price has on the other's demand.

```
* Merger counterfactual equilibrium prices
gen p_merge = p
forvalues i = 1/10000 {
        replace p_merge = p_merge + marg_profit
        drop s_denominator
        bysort market: egen s_denominator = total(exp(gamma - alpha*p_merge + xi))
        replace s_denominator = s_denominator + 1
        replace s = exp(gamma - alpha*p_merge + xi) / s_denominator
        sort market product
        replace marg_profit = s + (p_merge-mc)*(-alpha*s*(1-s)) ///
                if product == 2
        replace marg_profit = s + (p_merge-mc)*(-alpha*s*(1-s)) + ///
                (p_merge[_n+2] - mc[_n+2])*(alpha*s*s[_n+2]) if product == 1
        replace marg_profit = s + (p_merge-mc)*(-alpha*s*(1-s)) + ///
                (p_merge[_n-2] - mc[_n-2])*(alpha*s*s[_n-2]) if product == 3
}

bysort product: summarize p
bysort product: summarize p_merge
```

Below I present the average prices before the merger (`p`) and after the merger (`p_merge`).

```
. bysort product: summarize p

-------------------------------------------------------------------------------
-> product = 1

    Variable |        Obs        Mean    Std. dev.       Min        Max
-------------+---------------------------------------------------------
           p |      2,000    2.769566    .5505068    1.69063   4.551873

-------------------------------------------------------------------------------
-> product = 2

    Variable |        Obs        Mean    Std. dev.       Min        Max
-------------+---------------------------------------------------------
           p |      2,000    3.383275    .4343537   2.678478    4.82484

-------------------------------------------------------------------------------
-> product = 3

    Variable |        Obs        Mean    Std. dev.       Min        Max
-------------+---------------------------------------------------------
           p |      2,000     4.28657    .3623045   3.668805   6.000759

. bysort product: summarize p_merge

-------------------------------------------------------------------------------
-> product = 1

    Variable |        Obs        Mean    Std. dev.       Min        Max
-------------+---------------------------------------------------------
     p_merge |      2,000    2.846945    .5878798   1.694769   4.997261

-------------------------------------------------------------------------------
-> product = 2

    Variable |        Obs        Mean    Std. dev.       Min        Max
-------------+---------------------------------------------------------
     p_merge |      2,000    3.384334    .4343825   2.678545   4.825495

-------------------------------------------------------------------------------
-> product = 3

    Variable |        Obs        Mean    Std. dev.       Min        Max
-------------+---------------------------------------------------------
     p_merge |      2,000    4.390863    .4568421   3.673826   6.953452
```

Estimated Merger Simulation

Next we try to calculate the counterfactual post-merger prices using only variables that are observable to the expnometrician. In particular, prices, quantities and distance to production site. First make sure you have the `mergersim` package installed in Stata.

We load the appropriate dataset and initialize `mergersim`. Because quantities are defined as market shares, we set the total market size to 1 for all markets.

```
clear all
set seed 12345
cd "C:\Users\marcu\Dropbox\Demand Est\Seminar 2"
use "1_data\seminar_2_data_observable.dta"

gen msize = 1

xtset product market

mergersim init, price(p) quantity(s) ///
        marketsize(msize) firm(firm)
```

Recall from slides that with logit demand:

$$\log(s_{jt}) - \log(s_{0t}) = \gamma_j - \alpha p_{jt} + \xi_{jt}$$

When we initialize `mergersim`, the left hand side expression here is automatically generated as `M_ls`.

Since p_{jt} is endogenous (correlated with ξ_{jt}), OLS yields a biased coefficient of $-\alpha$. Since $CoV(p_{jt}, \xi_{jt}) > 0$, we expect that $-\hat{\alpha}_{OLS} > -\alpha = -1.5$. Indeed:

```
* Price is endogenous so OLS is biased
reg M_ls p i.product
```

```
. * Price is endogenous so OLS is biased
. reg M_ls p i.product

      Source |       SS           df       MS      Number of obs   =     6,000
-------------+----------------------------------   F(3, 5996)      =     18.38
       Model |  199.451806         3  66.4839354   Prob > F        =    0.0000
    Residual |  21691.5485     5,996  3.61766987   R-squared       =    0.0091
-------------+----------------------------------   Adj R-squared   =    0.0086
       Total |  21891.0003     5,999  3.64910824   Root MSE        =     1.902

------------------------------------------------------------------------------
        M_ls | Coefficient  Std. err.      t    P>|t|     [95% conf. interval]
-------------+----------------------------------------------------------------
           p |  -.2911113   .0538976    -5.40   0.000    -.3967699   -.1854527
             |
     product |
          2  |   .1931612   .0686424     2.81   0.005     .0585973    .3277251
          3  |   .1837695   .1015028     1.81   0.070    -.0152125    .3827516
             |
       _cons |  -2.311305   .1552135   -14.89   0.000     -2.61558   -2.007031
------------------------------------------------------------------------------
```

To resolve the endogeneity issue, we use an instrumental variable. The instrument we use is *distance to production site*. This instrument is both relavant and exogenous in our model:

- **Relevance:** Marginal cost is increasing in distance to production site. Moreover, marginal cost leads to higher price in this model. Note that in reality, firms may charge the same price in locations with different marginal cost.
- **Exclusion restriction:** By assumption, consumer tastes are uncorrelated with distance to prodcution sites. If this was not the case (e.g. if people who live closer to where a product is produced tend to have a preference for that product), then the exclusion restriction would be violated and the IV estimator would be biased.

```
* Use distance to production site as an instrument for price (cost shifter)
ivregress 2sls M_ls (p = dist_to_production) i.product
```

```
. ivregress 2sls M_ls (p = dist_to_production) i.product

Instrumental variables 2SLS regression            Number of obs   =      6,000
                                                  Wald chi2(3)    =     611.66
                                                  Prob > chi2     =     0.0000
                                                  R-squared       =          .
                                                  Root MSE        =      1.976

------------------------------------------------------------------------------
        M_ls | Coefficient  Std. err.      z    P>|z|     [95% conf. interval]
-------------+----------------------------------------------------------------
           p |  -1.471627    .060709   -24.24   0.000    -1.590615    -1.35264
             |
     product |
          2  |   .9176544   .0727506    12.61   0.000     .7750659    1.060243
          3  |   1.974617   .1112931    17.74   0.000     1.756486    2.192747
             |
       _cons |   .9582117   .1738463     5.51   0.000     .6174792    1.298944
------------------------------------------------------------------------------
Instrumented: p
 Instruments: 2.product 3.product dist_to_production
```

As we see, the estimate here of -1.472 is very close to the true $-\alpha$ of -1.5.

Finally, we estimate the pre-merger conditions (marginal cost and Lerner index) and simulate a merger between firm 1 and 3.

```
* Pre-Merger conditions
mergersim market

* Merger Simulation
mergersim simulate, buyer(1) seller(3) method(fixedpoint) maxit(40) dampen(0.5) detail
```

```
  Pre-merger Market Conditions
  Unweighted averages by firm

  ----------------------------------------------------------------------
           firm |          p     Marginal costs     Pre-merger Lerner
  --------------+-------------------------------------------------------
              1 |      2.770              1.985                 0.295
              2 |      3.383              2.610                 0.231
              3 |      4.287              3.530                 0.177
  ----------------------------------------------------------------------
  Variables generated: M_costs M_delta
  ----------------------------------------------------------------------
```

Prices
Unweighted averages by firm

firm	Pre-merger	Post-merger	Relative change
1	**2.770**	**2.849**	**0.028**
2	**3.383**	**3.384**	**0.000**
3	**4.287**	**4.394**	**0.024**

Variables generated: **M_price2 M_quantity2 M_price_ch (Other M_ variables dropped)**

In summary, the underlying priors (marginal cost and demand system) can be precisely estimated and the post-merger price can be simulated using these. In this case, the empirical model accurately predicts that mean prices of product 1 and 3 would rise by about 2.8% and 2.4% respectively while that of product 2 would remain virtually unchanged.

Object	Product	True	Estimated
α	-	1.5	1.471
Mean Marginal Cost	1	2.0005	1.985
	2	2.62525	2.610
	3	3.54485	3.530
Mean Pre-Merger Price	1	2.770	
	2	3.383	
	3	4.287	
Mean Post-Merger Price	1	2.846945	2.849
	2	3.384334	3.384
	3	4.390863	4.394